<!--#include file="lib/utf-8.asp"-->
<!--#include file="lib/connopen.asp"-->
<!--#include file="lib/asp-common-lib.asp"-->
<%

'获取提交查询的参数
tx_search=trim(SafeRequest("tx_search",0))
class_id=trim(SafeRequest("class_id",1))
vendor_id=trim(SafeRequest("vendor_id",1))
search_count=trim(SafeRequest("search_count",1))
stock_alarm=trim(SafeRequest("stock_alarm",0))
'组合sql查询字符串
'sql="select * from cdfee where true "
sql="select "
if search_count<>0 then sql=sql&" top "&search_count
sql=sql&" * from stock where true "
if tx_search<>"" then
    sql=sql&" and (goods_name like '%"&tx_search&"%' or goods_model like '%"&tx_search&"%' or goods_memo like '%"&tx_search&"%' or vendor_name like '%"&tx_search&"%' or brand_name like '%"&tx_search&"%' or class_name like '%"&tx_search&"%')"
end if
if class_id<>0 then sql=sql&" and class_id="&class_id
if vendor_id<>0 then sql=sql&" and vendor_id="&vendor_id
if stock_alarm<>"" then 
    if stock_alarm=0 then
        sql=sql&" and goods_stock/goods_quantity>0"
    else
        sql=sql&" and goods_stock/goods_quantity<"&stock_alarm
    end if
end if
sql=sql&" order by class_name asc,goods_name asc"
' response.write sql
'response.end
'执行数据查询
set rs = Server.CreateObject("ADODB.Recordset")
rs.open sql,conn,1,1
If rs.eof Then
  response.write("没有检索到数据!")
  response.end
End if

'分页控制
If saferequest("page",0)<>"" Then  page=clng(saferequest("page",0)) Else  page=1
pc=saferequest("pc",1)
If pc="" Then pc=25
rs.pagesize=pc
pagesize=rs.pagesize
pagecount=rs.pagecount
'page=clng(request.QueryString("page"))
if Not IsEmpty(page) then '如果page_num已经初始化...
  if Not IsNumeric(page) then '判断page_num值是否为数字
    page=1
  else
    page = clng(page) '接收page_num并化为数字型赋给page_num变量
  end if
  if page > pagecount then '如果接收的页数大于总页数
    rs.AbsolutePage = pagecount '设置当前显示页等于最后页
  elseif page <= 0 then '如果page小于等于0
    page=1
    rs.AbsolutePage = page '设置当前显示页等于第一页
  else
    rs.AbsolutePage = page '如果大于零,显示当前页等于接收的页数
  end if
  page = rs.AbsolutePage
end if

%>



<table class="full">
  <thead>
    <tr>
      <th scope="col">序号</th>
      <th scope="col">文具批号</th>
      <th scope="col">分类</th>
      <th scope="col">文具名</th>
      <th scope="col">单位</th>
      <th scope="col">型号</th>
      <th scope="col">规格</th>
      <th scope="col">品牌</th>
      <th scope="col">单价</th>
      <th scope="col">进货量</th>
      <th scope="col">库存</th>
      <th scope="col">进货日期</th>
      <th scope="col">备注</th>
    </tr>
  </thead>
  <tbody>
<%
  For i=1 To pagesize
    If rs.eof Then Exit for
%>
    <tr>
      <th class="sub" scope="row"><%=rs.AbsolutePosition%></th>
      <td><%=rs("goods_purchase")%></td>
      <td><%=rs("class_name")%></td>
      <td><a href="goods.asp?goods_id=<%=rs("goods_id")%>"><%=rs("goods_name")%></a></td>
      <td><%=rs("unit_name")%></td>
      <td><%=rs("goods_model")%></td>
      <td><%=rs("goods_spec")%></td>
      <td><%=rs("brand_name")%></td>
      <td><%=rs("goods_unitprice")%></td>
      <td><%=rs("goods_quantity")%></td>
      <td><%=rs("goods_stock")%></td>
      <td><%=rs("purchase_date")%></td>
      <td><%=rs("goods_memo")%></td>
    </tr>
<%
  rs.movenext()
next
%>
  </tbody>
</table>
<div>
  <span id="pagebar">
    共查询到&nbsp;<%=rs.RecordCount%>&nbsp;条记录&nbsp;
    <%  If page>1 Then  %>
      [<a href="#" id="1">首页</a>]
      [<a href="#" id="<%=page-1%>">上一页</a>]
    <%
        End If
        If Not (page>=pagecount Or pagecount=0) Then
    %>
      [<a href="#" id="<%=page+1%>">下一页</a>]
      [<a href="#" id="<%=pagecount%>">尾页</a>]
    <%  End If %>
      第 <%=page%> 页 / 共 <%=pagecount%> 页
      转到<input id="page_val" type="text" size="2" maxlength="<%=Int(Log(pagecount))+1%>" value="<%=page%>" />页
      <input id="page_go" class="bt_button" type="button" value="Go" />
  </span>
</div>
<%
rs.close
Set rs=nothing
%>
<!--#include file="lib/connclose.asp"-->

